

Polymorphism and type classes


Haskell and Cryptocurrencies

Dr. Andres Löb, Well-Typed LLP

Dr. Lars Brünjes, IOHK

Alejandro Garcia, IOHK

2020-07-24



INPUT | OUTPUT

Type inference

- The compiler will infer types for expressions, and for constant and function declarations automatically. Type annotations are rarely required.
- Type annotations can always be provided and will be checked for correctness by the compiler.
- Type signatures for top-level declarations are considered good style. They serve as invaluable machine-checked interface documentation.
- You can use GHC(i) to obtain inferred types. Use `:t` often, but also try to train your own type inference capabilities over time – it will help you to understand errors with less effort.

Parametric polymorphism

One function, several types

Some Haskell expressions and functions can have more than one type.

Example:

```
fst (x, y) = x
```

Possible type signatures (all would work):

```
fst :: (a, a) -> a  
fst :: (Int, a) -> Int  
fst :: (Int, Int) -> Int  
fst :: (a, b) -> a  
fst :: (Int, Char) -> Int
```

Is one of these clearly the “best” choice?

Most general type

Haskell's type system is designed such that (ignoring some language extensions) each term has a *most general type*:

- the most general type allows the most flexible use;
- all other types the term has can be obtained by instantiating the most general type, i.e., by substituting type variables with type expressions.

Instantiating types

The type signature

```
fst :: (a, b) -> a
```

declares the most general type for `fst`. Types like

```
fst :: (a, a) -> a
```

```
fst :: (Int, Char) -> Int
```

```
fst :: (a -> Int -> b, c) -> a -> Int -> b
```

are instantiations of the most general type.

Instantiating types

The type signature

```
fst :: (a, b) -> a
```

declares the most general type for `fst`. Types like

```
fst :: (a, a) -> a
```

```
fst :: (Int, Char) -> Int
```

```
fst :: (a -> Int -> b, c) -> a -> Int -> b
```

are instantiations of the most general type.

Type inference will always infer the most general type!

No run-time type information

Haskell terms carry no type information at run-time.

Remember

You can only ever use a term in the ways its type dictates.

No run-time type information

Haskell terms carry no type information at run-time.

Remember

You can only ever use a term in the ways its type dictates.

Example:

```
fst :: (a, b) -> a
fst (x, y) = x

restrictedFst :: (Int, Int) -> Int
restrictedFst = fst  -- ok
```

```
newFst :: (a, b) -> a
newFst = restrictedFst  -- type error!
```

Parametric polymorphism

- A type with type variables (but no class constraints) is called (*parametrically*) *polymorphic*.
- Type variables can be instantiated to any type expression, but several occurrences of the same variable have to be the same type.
- If a function argument has polymorphic type, then you know nothing about it. No pattern matching is possible. You can only pass it on.
- If a function result has polymorphic type, then (except for **undefined** and **error**) you can only try to build one from the function arguments.

Let us look at examples.

Example

How many functions can you think of that have this type:

```
(Int, Int) -> (Int, Int)
```

Example

How many functions can you think of that have this type:

```
(Int, Int) -> (Int, Int)
```

And of this one?

```
(a, a) -> (a, a)
```

Example

How many functions can you think of that have this type:

```
(Int, Int) -> (Int, Int)
```

And of this one?

```
(a, a) -> (a, a)
```

And of this one?

```
(a, b) -> (b, a)
```

(Thanks to Doaitse Swierstra for the example.)

Parametricity

- In general, parametric polymorphism severely restricts how a function can be implemented.
- So if the functionality you're trying to implement is quite general, this is a good thing, because it really prevents you from making errors.
- Conversely, if you see a function with parametrically polymorphic type, you *know* that it cannot look at the polymorphic values.
- By looking at polymorphic types alone, one can obtain non-trivial properties of the functions. (This is sometimes called “parametricity”.)

Parametricity for `map`

```
map :: (a -> b) -> [a] -> [b]
```

```
map :: (a -> b) -> [a] -> [b]
```

Must produce a list in which all elements are obtained by applying the given function to elements of the original list.


```
map :: (a -> b) -> [a] -> [b]
```

Must produce a list in which all elements are obtained by applying the given function to elements of the original list.

But we don't know how long the resulting list is, or in which order the elements occur.

A common pitfall: who gets to choose

Sometimes, it may be tempting to write a program like the following:

```
parse :: String -> a
parse "False" = False
parse "0"      = 0
...
```

What is wrong here?

A common pitfall: who gets to choose

Sometimes, it may be tempting to write a program like the following:

```
parse :: String -> a
parse "False" = False
parse "0"      = 0
...
```

What is wrong here?

For polymorphic types, it is always the caller who gets to choose at which type the function should be used.

A function with polymorphic result type (but no polymorphic arguments) is impossible to write without either looping or causing an exception: we'd have to produce a value that belongs to every type imaginable!

What if we need to return values of different types?

Option 1: use `Either`:

```
data Either a b = Left a | Right b
parse :: String -> Either Bool Int
parse "False" = Left False
parse "0"      = Right 0
```

What if we need to return values of different types?

Option 1: use `Either`:

```
data Either a b = Left a | Right b
parse :: String -> Either Bool Int
parse "False" = Left False
parse "0"      = Right 0
```

Option 2: define your own datatype.

```
data Value = Value Bool | VInt Int
parse :: String -> Value
parse "False" = Value False
parse "0"     = VInt 0
```

The second option is quite common in libraries that interface with dynamically typed languages (SQL, JSON, ...).

Overloading

Reusing code, reusing names

Parametric polymorphism

Allows you to use the same implementation in as many contexts as possible.

Overloading (ad-hoc polymorphism)

Allows you to use the same function name in different contexts, but with different implementations for different types.

Type classes

A *type class* defines an interface that can be implemented by potentially many different types.

Example:

```
class Eq a where  
  (==) :: a -> a -> Bool  
  (/=) :: a -> a -> Bool
```


Type classes

A *type class* defines an interface that can be implemented by potentially many different types.

Example:

```
class Eq a where
  (==) :: a -> a -> Bool
  (/=) :: a -> a -> Bool
```

Using **instance** declarations, we can explain how a certain type (or types of a certain shape) implement the interface.

Instances

```
instance Eq Bool where
  False == False = True
  True   == True   = True
  _       == _       = False
  x /= y = not (x == y)
```

```
instance Eq a => Eq [a] where
  []      == []      = True
  (x : xs) == (y : ys) = x == y && xs == ys
  _        == _        = False
  xs /= ys = not (xs == ys)
```

Instances

```
instance Eq Bool where
  False == False = True
  True   == True   = True
  _      == _      = False
  x /= y = not (x == y)
```

```
instance Eq a => Eq [a] where
  []      == []      = True
  (x : xs) == (y : ys) = x == y && xs == ys
  _       == _       = False
  xs /= ys = not (xs == ys)
```

We use equality on `a` while defining equality on `[a]`.

Class constraints

All the instances of a given type class specify a subset of all the Haskell types, namely the subset that implements the class interface.

Class constraints

All the instances of a given type class specify a subset of all the Haskell types, namely the subset that implements the class interface.

In type signatures, class constraints specify that a type variable can only be instantiated to types belonging to a certain class:

```
(==) :: Eq a => a -> a -> Bool
```

Read: “Given that `a` is an instance of `Eq`, the function has the type `a -> a -> Bool`.”

Overloading and inference

Not only class methods, but also functions that directly or indirectly use class methods can have types with constraints.

Example:

```
allEqual :: Eq a => [a] -> Bool
allEqual []          = True
allEqual [x]         = True
allEqual (x : y : ys) = x == y && allEqual (y : ys)
```

Also recall `elem` or `lookup`.

Class constraints will be automatically inferred by the compiler.

Several class constraints

There can be multiple constraints on a function, and they can apply to several variables:

```
example ::  
  ... => (a, b) -> (a, b) -> String  
example (x1, y1) (x2, y2)  
  | x1 == x2 && y1 == y2 = show x1  
  | otherwise           = "different"
```

Can you infer the constraints?

Several class constraints

There can be multiple constraints on a function, and they can apply to several variables:

```
example ::  
  (Eq a, Eq b, Show a) => (a, b) -> (a, b) -> String  
example (x1, y1) (x2, y2)  
  | x1 == x2 && y1 == y2 = show x1  
  | otherwise            = "different"
```


Default definitions

```
class Eq a where
  (==) :: a -> a -> Bool
  (/=) :: a -> a -> Bool
  x == y = not (x /= y)
  x /= y = not (x == y)
```

Now:

```
instance Eq Bool where
  False == False = True
  True   == True  = True
  _      == _     = False
```

And `(/=)` will work automatically.

Minimal class implementations

Providing *neither* `(==)` nor `(/=)` produces a compiler warning, not an error!

Minimal class implementations

Providing *neither* `(==)` nor `(/=)` produces a compiler warning, not an error!

In this case, it leads to non-terminating (mutually recursive) definitions of both functions!

Minimal class implementations

Providing *neither* `(==)` nor `(/=)` produces a compiler warning, not an error!

In this case, it leads to non-terminating (mutually recursive) definitions of both functions!

There are **MINIMAL** compiler pragmas that can be used to explain which combinations of methods should be implemented. These are displayed in GHCi when `:i` is used on a class.

Classes are not types!

Note that

```
f :: Eq -> Eq -> Bool
```

```
f :: Eq a -> Eq a -> Bool
```

are both invalid. Classes appear in constraints!

Classes are not types!

Note that

```
f :: Eq -> Eq -> Bool
```

```
f :: Eq a -> Eq a -> Bool
```

are both invalid. Classes appear in constraints!

Also note that the type

```
Eq a => a -> a -> Bool
```

forces both arguments to be of the same type. You cannot pass two different types that are both an instance of `Eq` – that would require a function of type

```
(Eq a, Eq b) => a -> b -> Bool
```

Important classes

- For equality and inequality.
- Note that equality in Haskell is structural equality. There is no “object identity”, and no pointer equality.
- Supported by most datatypes, such as numbers, characters, tuples, lists, `Maybe`, `Either`, ...
- Not supported for function types.

For comparisons between values of the same type.

```
class Eq a => Ord a where
  compare :: a -> a -> Ordering
  (<)      :: a -> a -> Bool
  (<=)     :: a -> a -> Bool
  (>)      :: a -> a -> Bool
  (>=)     :: a -> a -> Bool
  max      :: a -> a -> a
  min      :: a -> a -> a
```

Several default definitions – you'd typically define just

`compare` or `(<=)`.

```
data Ordering = LT | EQ | GT
```

Superclasses

```
class Eq a => Ord a where  
  ...
```

The condition indicates that `Eq` is a *superclass* of `Ord`:

- You cannot give an instance for `Ord` without first providing an instance to `Eq`.
- Conversely, a constraint `Ord a => ...` on a function implies `Eq a`. In other words, `(Ord a, Eq a) => ...` is equivalent to `Ord a => ...`.

```
class Show a where
  show      :: a -> String
  showsPrec :: Int -> a -> ShowS
  showList  :: [a] -> ShowS
```

The most important method is `show`:

- used to produce a human-readable `String`-representation of a value;
- it is sufficient to define `show` in new instances, as the others have default definitions;
- the other two functions can be used to more efficiently and beautifully implement `show` internally (for example, remove unnecessary parentheses).

The `Show` class is also used by GHCi to print result values of evaluated terms.

The `Show` class is also used by GHCi to print result values of evaluated terms.

If you evaluate an expression of a type that has no `Show` instance, you will get an error in GHCi complaining about a missing `Show` constraint – this does *not* indicate an actual programming error.

Read

```
class Read a where  
  readsPrec :: Int -> ReadS a  
  readList  :: ReadS [a]
```

Most often, the derived function `read` is used:

```
read :: Read a => String -> a
```

Tries to interpret a given `String` (such as produced by `show`) as a value of a type.

Read

```
class Read a where
  readsPrec :: Int -> ReadS a
  readList  :: ReadS [a]
```

Most often, the derived function `read` is used:

```
read :: Read a => String -> a
```

Tries to interpret a given `String` (such as produced by `show`) as a value of a type.

How the value is interpreted is statically determined by the context:

```
read "1" + 2      -- used and parsed as a number
not (read "False") -- used and parsed as a Bool
```

Unresolved overloading

The following function produces an error (not in GHCi, but if placed in a file):

```
strange x = show (read x)
```

The error will say something about an “ambiguous type variable” and mention constraints for `Read` and `Show`.

Can you imagine what the problem is?

Unresolved overloading

The following function produces an error (not in GHCi, but if placed in a file):

```
strange x = show (read x)
```

The error will say something about an “ambiguous type variable” and mention constraints for `Read` and `Show`.

Can you imagine what the problem is?

The `x` is a `String` which is then parsed into something by `read`. But what type should it be parsed at? The context does not tell, because the result is passed to `Show`, which is also overloaded.

Manually resolving overloading

This works:

```
strange :: String -> String  
strange x = show (read x :: Bool)
```

Or this:

```
strange :: String -> String  
strange x = show (read x :: Int)
```

But note that the choice of intermediate type does make a difference!

Manually resolving overloading

This works:

```
strange :: String -> String  
strange x = show (read x :: Bool)
```

Or this:

```
strange :: String -> String  
strange x = show (read x :: Int)
```

But note that the choice of intermediate type does make a difference!

In general, if several overloaded functions are combined such that the resulting type does not mention any overloaded variables anymore, you have to specify the intermediate types manually to help the type checker resolve the overloading.

deriving

For a limited number of type classes (but in particular `Eq`, `Ord`, `Show`, `Read`), the Haskell compiler has a built-in algorithm to derive an instance for nearly any datatype.

deriving

For a limited number of type classes (but in particular `Eq`, `Ord`, `Show`, `Read`), the Haskell compiler has a built-in algorithm to derive an instance for nearly any datatype.

```
data Tree a = Leaf a | Node (Tree a) (Tree a)
  deriving (Eq, Ord, Show, Read)
```

Defines the `Tree` datatype of binary trees together with suitable instances:

- equality is always deep and structural;
- ordering depends on the order of constructors;
- `Show` and `Read` assume the natural human-readable Haskell string representation.

Numbers

Numeric types and classes

There are several numeric types and classes in Haskell:

<i>type</i>	<i>instance of</i>		
<code>Int</code>	<code>Num</code>	<code>Integral</code>	
<code>Integer</code>	<code>Num</code>	<code>Integral</code>	
<code>Float</code>	<code>Num</code>	<code>Fractional</code>	<code>Floating</code> <code>RealFrac</code>
<code>Double</code>	<code>Num</code>	<code>Fractional</code>	<code>Floating</code> <code>RealFrac</code>
<code>Rational</code>	<code>Num</code>	<code>Fractional</code>	

Numeric types and classes

There are several numeric types and classes in Haskell:

<i>type</i>	<i>instance of</i>		
<code>Int</code>	<code>Num</code>	<code>Integral</code>	
<code>Integer</code>	<code>Num</code>	<code>Integral</code>	
<code>Float</code>	<code>Num</code>	<code>Fractional</code>	<code>Floating</code> <code>RealFrac</code>
<code>Double</code>	<code>Num</code>	<code>Fractional</code>	<code>Floating</code> <code>RealFrac</code>
<code>Rational</code>	<code>Num</code>	<code>Fractional</code>	

The class `Num` is a *superclass* of `Integral`.

The class `Fractional` is a superclass of `Floating`.

Numeric types and classes

There are several numeric types and classes in Haskell:

<i>type</i>	<i>instance of</i>	
<code>Int</code>	<code>Num</code>	<code>Integral</code>
<code>Integer</code>	<code>Num</code>	<code>Integral</code>
<code>Float</code>	<code>Num</code>	<code>Fractional Floating RealFrac</code>
<code>Double</code>	<code>Num</code>	<code>Fractional Floating RealFrac</code>
<code>Rational</code>	<code>Num</code>	<code>Fractional</code>

The class `Num` is a *superclass* of `Integral`.

The class `Fractional` is a superclass of `Floating`.

- Whereas `Int` is bounded, `Integer` is unbounded (bounded by memory only).
- A `Double` is usually of higher precision than a `Float`.
- The datatype `Rational` is for fractions.

Operations on numbers

Most operations on numbers and even numeric literals are *overloaded*:

```
(+) :: (Num a) => a -> a -> a
```

```
(-) :: (Num a) => a -> a -> a
```

```
(*) :: (Num a) => a -> a -> a
```

Operations on numbers

Most operations on numbers and even numeric literals are *overloaded*:

```
(+) :: (Num a) => a -> a -> a  
(-) :: (Num a) => a -> a -> a  
(*) :: (Num a) => a -> a -> a
```

```
1    :: (Num a) => a    -- overloaded literals  
1.2  :: (Fractional a) => a  -- overloaded literals
```

Operations on numbers

Most operations on numbers and even numeric literals are *overloaded*:

```
(+) :: (Num a) => a -> a -> a  
(-) :: (Num a) => a -> a -> a  
(*) :: (Num a) => a -> a -> a
```

```
1    :: (Num a) => a    -- overloaded literals  
1.2  :: (Fractional a) => a  -- overloaded literals
```

```
(/) :: (Fractional a) => a -> a -> a  
mod :: (Integral a) => a -> a -> a  
div :: (Integral a) => a -> a -> a  
sin :: (Floating a) => a -> a  
log :: (Floating a) => a -> a
```

No automatic coercion

We can use overloaded functions at different types:

```
3 * 4
```

```
3.2 * 4.5
```

But there is no implicit coercion:

```
3.2 * (5 `div` 2) -- type error
```

```
3.2 * fromIntegral (5 `div` 2)
```

No automatic coercion

We can use overloaded functions at different types:

```
3 * 4  
3.2 * 4.5
```

But there is no implicit coercion:

```
3.2 * (5 `div` 2) -- type error  
3.2 * fromIntegral (5 `div` 2)
```

Question

Why is `3.2 * 2` ok, but not `3.2 * (5 `div` 2)`?

No automatic coercion

We can use overloaded functions at different types:

```
3 * 4  
3.2 * 4.5
```

But there is no implicit coercion:

```
3.2 * (5 `div` 2) -- type error  
3.2 * fromIntegral (5 `div` 2)
```

Question

Why is `3.2 * 2` ok, but not `3.2 * (5 `div` 2)`?

Because `2 :: (Num a) => a`, but

```
(5 `div` 2) :: (Integral a) => a
```

Converting between numeric types

From an integral type to another:

```
fromIntegral :: (Integral a, Num b) => a -> b
```

From a fractional type to an integral:

```
round    :: (RealFrac a, Integral b) => a -> b  
floor    :: (RealFrac a, Integral b) => a -> b  
ceiling  :: (RealFrac a, Integral b) => a -> b
```

Here, `round` rounds to the nearest even number.

Recap: Haskell types

Question

Which Haskell types have we seen so far?

Recap: Haskell types

Question

Which Haskell types have we seen so far?

Note that there are:

- truly built-in types such as `Int`, `Char` or functions `(->)`;
- types that could be defined by `data`, but support special syntax, such as tuples and lists;
- types that are defined in the basic libraries, but could just as well have been defined by you (`Bool`, `Maybe`, `Either`);
- types that are actually just synonyms for other types (`String`).

Use GHCi for information

A quick way to remind yourself of the definition of a datatype or class is by using `:i` or `:info` in GHCi:

- Shows the `data` declaration and class instances for datatypes.
- Shows the `class` declaration and instances for classes.
- Shows a (partial) `data` declaration for constructors.
- Shows a (partial) `class` declaration for methods.
- Shows the type signature for functions and constants.

Use GHCi for information

A quick way to remind yourself of the definition of a datatype or class is by using `:i` or `:info` in GHCi:

- Shows the `data` declaration and class instances for datatypes.
- Shows the `class` declaration and instances for classes.
- Shows a (partial) `data` declaration for constructors.
- Shows a (partial) `class` declaration for methods.
- Shows the type signature for functions and constants.

Sometimes, GHCi lets you glimpse at internal implementation details that are – at this point – difficult to understand (such as for `:i Int`).

Higher-order functions

Functions, functions, functions

A function parameterized by another function or returning a function is called a *higher-order function*.

Functions, functions, functions

A function parameterized by another function or returning a function is called a *higher-order function*.

Currying

Strictly speaking, every curried function in Haskell is a function returning another function:

```
elem :: Eq a => a -> ([a] -> Bool)
```

```
elem 3 :: (Eq a, Num a) => [a] -> Bool
```

Filtering and mapping

Two of the most useful list functions are higher-order, as they each take a function as an argument:

```
filter :: (a -> Bool) -> [a] -> [a]  
map    :: (a -> b) -> [a] -> [b]
```


Filtering and mapping

Two of the most useful list functions are higher-order, as they each take a function as an argument:

```
filter :: (a -> Bool) -> ([a] -> [a])  
map    :: (a -> b) -> ([a] -> [b])
```

The use of a function `a -> Bool` to express a predicate is generally common. And mapping a function over a data structure is an operation that isn't limited to lists.

Overloading vs. parameterization

Consider:

```
sort    :: Ord a          => [a] -> [a]  
sortBy :: (a -> a -> Ordering) -> [a] -> [a]
```

Overloading vs. parameterization

Consider:

```
sort    :: Ord a          => [a] -> [a]
sortBy :: (a -> a -> Ordering) -> [a] -> [a]
```

Both functions are rather similar:

- the first takes the comparison function to use from the **instance** declaration for the element type of the list;
- the second is passed an explicit comparison function.

Using an overloaded function is a bit more convenient, but using **sortBy** is a bit more flexible.

Overloading vs. parameterization

Consider:

```
sort    :: Ord a          => [a] -> [a]
sortBy :: (a -> a -> Ordering) -> [a] -> [a]
```

Both functions are rather similar:

- the first takes the comparison function to use from the **instance** declaration for the element type of the list;
- the second is passed an explicit comparison function.

Using an overloaded function is a bit more convenient, but using **sortBy** is a bit more flexible.

Interestingly, GHC implements overloaded functions by passing type class “dictionaries” as additional arguments.

Performance impact of overloading

- You pay no price whatsoever for parametric polymorphism.
- Overloaded functions get extra arguments at runtime. There is a slight performance penalty for that.
- Only overloaded functions get extra arguments – remember that there is no general run-time type information!
- It is possible to instruct GHC to generate specialized versions for overloaded functions at particular types, thereby eliminating the run-time overhead.
- GHC also has a relatively aggressive inliner. Inlining overloaded functions can also remove the overhead, much like specialization.

Recap

- Parametric polymorphism allows to use the same code for many different types.
- It is completely unrestricted: the caller can choose how to instantiate each type variable.
- Overloading can be used to use different implementations via the same name, and the implementation chosen is determined by the type checker.
- Overloading can not only occur on function arguments, but also on result types.
- Overloaded functions can also be used to define common functionality in terms of just a small interface.